

Modeling Diffusion Resistors

Ken Kundert

Designer's Guide Consulting, Inc.

Version 1b, 20 September 2002 Describes a simple empirical three-terminal model for a nonlinear diffusion resistor that overcomes many of the problems inherent to the common two-terminal model.

Last updated on May 12, 2006. You can find the most recent version at www.designers-guide.org. Contact the author via e-mail at ken@designers-guide.com.

Permission to make copies, either paper or electronic, of this work for personal or classroom use is granted without fee provided that the copies are not made or distributed for profit or commercial advantage and that the copies are complete and unmodified. To distribute otherwise, to publish, to post on servers, or to distribute to lists, requires prior written permission.


1 Diffusion Resistors

The cross-section of a diffusion resistor is shown in Figure 1. It is a structure very similar to a MOSFET with the gate removed. Thus, rather than a 4 terminal device, it is a three terminal device. The third terminal is the substrate (t_0), which acts to modulate the channel resistance in the same way that substrate acts as a back-gate in a MOSFET. It can be modeled as a JFET where the substrate acts as the gate.

In this paper a very simple empirical model is developed that can be easily extracted. This model is not meant to compete against the more sophisticated physics-based models [3, 4], rather this three-terminal model is provided as an alternative to the simple two-terminal model that are commonly used. Trying to model a three-terminal component with a two-terminal model results in several problems that are neatly resolved by this model.

Neglecting any high frequency effects, and assuming normal operation, the current through the resistor in the normal mode of operation flows between terminals t_1 and t_2 . However it is a three terminal element in that the characteristics are dependent on the voltage at the substrate terminal t_0 . The voltage of the substrate acts to modulate the resistance between t_1 and t_2 .

FIGURE 1 Cross-section of a p-type diffusion resistor.



A simple model for the resistance of a diffusion resistor patterned after that from [1] is

$$r(v_{\text{avg}}) = (1 + c_r v_{\text{avg}}) r_0 \quad (1)$$

where

$$v_{\text{avg}} = \frac{v_1 + v_2}{2} - v_0 \quad (2)$$

is the average value of v_1 and v_2 relative to v_0 . In this model, c_r is the voltage coefficient of the normalized resistance. This model assumes that r_0 is measured at $v_{\text{avg}} = 0$. This is a simple model that provides a first-order variation in the resistance with respect to changes in the channel-to-substrate voltage, v_{avg} .

A resistance formulation such as (1) is not convenient for use in circuit simulators, which use nodal analysis for which it is more efficient to have models that compute current as a function of voltage. The model can be reformulated as a conductance

$$g(v_{\text{avg}}) = (1 + c_g v_{\text{avg}}) g_0, \quad (3)$$

where $g_0 = 1/r_0$ and c_g is the voltage coefficient of the normalized conductance.

From (3) the current through the resistor is

$$i(v_{\text{diff}}, v_{\text{abs}}) = g(v_{\text{avg}})v_{\text{diff}} = (1 + c_g v_{\text{avg}})g_0 v_{\text{diff}} \quad (4)$$

where

$$v_{\text{diff}} = v_1 - v_2. \quad (5)$$

This model is linear with respect to v_{diff} and nonlinear with a first-order dependence with respect to v_{avg} .

The coefficients for the resistance, c_r , and the conductance, c_g , can be related as follows.

$$g(v_{\text{avg}}) = \frac{1}{r(v_{\text{avg}})}, \quad (6)$$

$$g_0(1 + c_g v_{\text{avg}}) = \frac{1}{(1 + c_r v_{\text{avg}})r_0}, \quad (7)$$

$$(1 + c_g v_{\text{avg}}) = \frac{(1 - c_r v_{\text{avg}})}{(1 + c_r v_{\text{avg}})(1 - c_r v_{\text{avg}})}, \quad (8)$$

$$(1 + c_g v_{\text{avg}}) = \frac{(1 - c_r v_{\text{avg}})}{(1 - (c_r v_{\text{avg}})^2)}. \quad (9)$$

Assuming $(c_r v_{\text{avg}})^2 \ll 1$ gives

$$(1 + c_g v_{\text{avg}}) \approx (1 - c_r v_{\text{avg}}), \quad (10)$$

or

$$c_g \approx -c_r. \quad (11)$$

The models of (1) and (3) are somewhat different. One assumes a first-order dependence on v_{avg} in the resistance and the other assumes it in the conductance. For small v_{avg} the difference between the two models is small. For large $c_r v_{\text{avg}}$ it is not obvious which is preferred. Extensive measurements would be needed to determine which model better fits the behavior of the resistors. However, it would be unusual for $c_r v_{\text{avg}}$ to be large enough for this to be an issue, especially if you carefully extract the parameters for the model you are using. The model derived from (3) is preferred in this paper because it is known to be more efficient within the simulator.

The model so far ignores the dependence in the conductance due to v_{diff} . Generally, diffusion resistors are implemented with a symmetric structure, and so the odd order terms in the conductance would be negligible. Generalizing this model to add a second-order voltage coefficient for the v_{diff} term gives

$$i(v_{\text{diff}}, v_{\text{abs}}) = \left(1 + c_1 v_{\text{avg}} + \frac{c_2 v_{\text{diff}}^2}{3}\right)g_0 v_{\text{diff}}. \quad (12)$$

In this case the conductance becomes

$$g(v_{\text{diff}}, v_{\text{abs}}) = \frac{di(v_{\text{diff}}, v_{\text{abs}})}{dv_{\text{diff}}} = (1 + c_1 v_{\text{avg}} + c_2 v_{\text{diff}}^2) g_0, \quad (13)$$

where c_1 is the first-order coefficient for v_{avg} and c_2 is the second-order coefficient for v_{diff} .

2 Implementing the Resistor

Spectre's physical resistor (*phy_res*) directly implements the first-order model, but in doing so it requires that a *model* statement be used to specify the details of the nonlinear behavior. This is desirable if there are a large number of resistors that have the same voltage coefficient. However, if each resistor has its own value for the coefficient, it becomes tedious to enter an extra model statement for every resistor. In Listing 1 the model statement was combined with the resistor in a subcircuit. With this approach, a model for a diffusion resistor is instantiated with a single line. For example,

```
R1 (dp dn) diff_res r0=1K cg=0.001
```

instantiates a copy of the subcircuit of Listing 1 with $r_0 = 1 \text{ K}\Omega$ and $c_g = 0.001$. Knowing that diffusion resistors normally has diodes between the body and the channel, Spectre's physical resistor issues a warning if those diodes, which are not modeled in this case, would become forward biased. The *subtype=poly* was added to the model to suppress this behavior.

LISTING 1 *Spectre subcircuit for a first-order diffusion resistor model.*

```
subckt diff_res (1 2)
  parameters r0 cg=0 // r0 is resistance when v1 and v2 both equal 0
                    // cg is voltage coefficient for the normalized conductance
  R (1 2) diff_res_mod r=r0
  model diff_res_mod phy_res polyarg=sum coeffs=[cg] subtype=poly
ends diff_res
```

Notice that the subcircuit only has two terminals. The third terminal is implicitly connected to ground by the physical resistor *R*. This was done in order to allow the model to directly replace existing two-terminal models. Of course, this assumes that the voltage at t_0 is fixed, in this case to 0. If one wishes to model the effect of changes in the voltage at t_0 , then it is necessary to modify the subcircuit given to explicitly bring t_0 out as a third terminal on the subcircuit. With t_0 grounded, $v_0 = 0$ and so the effective resistance is $r = r_0$ when $v_{\text{avg}} = 0$.

The existing version of the physical resistor is not able to implement the second-order model, so a Verilog-A [2,5] model is provided instead. It is shown in Listing 2. One advantage that this model has over the previous one is that v_0 is available as a parameter that can be adjusted to give a better fit.

Be aware that these models are polynomial-based and so they can give non-physical results if either c is set much too large or they are used well outside of their intended operating region. This can lead to convergence difficulties and stability problems.

LISTING 2 Verilog-A description for a second-order diffusion model.

```

`include "discipline.h"
module diff_res (t1, t2);
  electrical t1, t2;

  parameter real r0=1K;    // resistance when v(t1)=v0 and v(t2)=v0
  parameter real c1=0;    // Coefficient of Vavg for normalized conductance
  parameter real c2=0;    // Coefficient of Vdiff^2 for normalized conductance
  parameter real v0=0;    // Back-gate voltage
  real Vdiff, Vavg, Geff;

  analog begin
    Vdiff = V(t1,t2);
    Vavg = (V(t1) + V(t2))/2 - v0;
    I(t1,t2) <+ (1 + c1*Vavg + c2*Vdiff*Vdiff/3)*Vdiff/r0;
    // Noise
    Geff = (1 + c1*Vavg + c2*Vdiff*Vdiff)/r0;
    I(t1,t2) <+ white_noise(4*P_K*$temperature*Geff);
  end
endmodule

```

3 Extracting the Model

A common approach to extracting the voltage coefficient for (3) or (13) is to connect t_2 and t_0 to ground and measure the conductance as a function of v_1 , the voltage applied to t_1 . Conductance is an incremental quantity,

$$g(v_1) = \frac{di_1}{dv_1}. \quad (14)$$

If the current is measured for finite number of steps spread evenly over a range of voltages, then the conductance can be approximated by

$$g(v_1) \approx \frac{\Delta i_1}{\Delta v_1}. \quad (15)$$

To extract the first-order model, assume that the conductance is a linear function of the v_1 , as it would be with (3),

$$g(v) = g_0 + g_1 v, \quad (16)$$

Then g_0 and g_1 can be extracted using linear regression on $g(v_1)$. Using (3) and the fact that for this test circuit $v_{\text{avg}} = v_1/2$,

$$c_g = 2 \frac{g_1}{g_0}. \quad (17)$$

To extract the second-order model, assume that the conductance is a quadratic function of the v_1 , as it would be with (13),

$$g(v) = g_0 + g_1 v + g_2 v^2, \quad (18)$$

Then g_0 , g_1 and g_2 are extracted by fitting this quadratic to the conductance data. Using (13) and the fact for this test circuit $v_{\text{avg}} = v_1/2$,

$$c_1 = 2 \frac{g_1}{g_0} \quad (19)$$

and

$$c_2 = \frac{g_2}{g_0}. \quad (20)$$

Alternatively, one can fit to the currents rather than the conductance. To extract c_g , one needs to fit a quadratic to the current data. To extract c_1 and c_2 one needs to fit a cubic. This is a more difficult procedure, but is expected to be better conditioned because the data is not differentiated before performing the fit.

4 Traditional Model

A common model for users of simulators such as HSPICE approximates the diffusion resistor using a two-terminal component that defines the resistance to be

$$r(v) = r_0(1 + c_1|v| + c_2|v|^2 + \dots). \quad (21)$$

There are two problems with this model. The first is that it is formulated in terms of resistance as a function of voltage. Remember that resistance is defined as the derivative voltage with respect to current. So that

$$r(v) = \frac{dv(v)}{di} = r_0(1 + c_1|v| + c_2|v|^2 + \dots). \quad (22)$$

The simulator needs to compute the current with respect to voltage, which is calculated from (22) with

$$i(v) = \int \frac{dv}{r_0(1 + c_1|v| + c_2|v|^2 + \dots)}. \quad (23)$$

This is a very difficult calculation to perform, and so an approximation is made. Instead of using the correct interpretation of resistance as $r = dv/di$, resistance is interpreted to be $r = v/i$, meaning that

$$i(v) = \frac{v}{r_0(1 + c_1|v| + c_2|v|^2 + \dots)}. \quad (24)$$


The problem is that this is incorrect if $c_k \neq 0$, and can be significantly in error if c_k is sufficiently different from 0. This particular problem is not especially troubling because there is only a significant difference in the $i(v)$'s computed by (23) and (24) for values of c_k that are considerably larger than one typically encounters in practice.

The second problem is more troubling. This model achieves the desired symmetry in a two terminal model by applying the absolute value to the odd order terms in the power series. This creates a noticeable kink in the IV characteristics of the resistor. This, in turn, tends to create considerable distortion that is both hard to ignore and not represen-

tative of the distortion that occurs in practice. The fact is that there is no kink in the actual characteristics of the resistor about $v = 0$. The kink present in the model causes significant amounts of energy to appear over a broad range of harmonics, and because the model is discontinuous, the energy in the harmonics decreases slowly with the size of the applied signal, meaning that this artificial distortion will likely dominate valid sources of distortion at low signal levels, which would invalidate such distortion results. This is particularly troubling when trying to measure asymptotic distortion metrics such as the intercept points (IP_3 , IP_5 , etc.) that must be measured at low signal levels.

This problem is quantified in Figures 2 and 3, which shows the level of the current at various harmonics when the resistor is driven with a pure sinusoidal voltage. Here it is assumed that $c_k = 0$ for all $k \neq 1$ and the power in the harmonics relative to the power in the fundamental is plotted as a function of $c_1 v$. Both the traditional model of (24) and the new model of (13) produce the expected level of 3rd-order harmonic distortion. However, the traditional model also generates significant amounts of distortion at all odd-order harmonics, where the new model generates none (measured to be over 300 dB down from the fundamental).

FIGURE 2 Normalized distortion versus $c_1 v$ for the traditional model of (24).



To understand the impact of this issue in practice, consider the circuit shown in Figure 4. This is a simple differential pair with emitter degeneration. The degeneration is performed using a diffusion resistor. The circuit was simulated with a 250 mV differential input and the spectrum of the differential output current is shown in Table 1. These results have been normalized to the power in the fundamental. The differences are especially pronounced in the higher harmonics stem from the kink that results from using the absolute value to create symmetric behavior in a two terminal model in (24).

5 If You Have Questions

If you have questions about what you have just read, feel free to post them on the *Forum* section of *The Designer's Guide Community* website. Do so by going to www.designers-guide.org/Forum.

FIGURE 3 Normalized distortion versus c_1v for the the new model of (13).




FIGURE 4 Schematic for a simple yet representative test circuit.




TABLE 1 Distortion produced by the circuit of Figure 4 using different models for the degeneration resistor.

Harmonic	Traditional Model	New Model
1	0.000 dB	0.000 dB
3	-53.905 dB	-54.876 dB
5	-91.849 dB	-85.990 dB
7	-99.959 dB	-116.077 dB
9	-108.489 dB	-145.852 dB

References

- [1] Akira Ito. Modeling of voltage-dependent diffused resistors. *IEEE Transactions on Electron Devices*, vol. 44, no. 12, Dec. 1997, pp. 2300-2302.
- [2] Kenneth S. Kundert. *The Designer's Guide to Verilog-AMS*. Kluwer Academic Publishers, 2004.

- [3] C. C. McAndrew, S. Sekine, A. Cassagnes, and Zhicheng Wu. Physically-based effective width modeling of MOSFETs and diffused resistors. *Proceedings of the 2000 International Conference on Microelectronic Test Structures (ICMTS 2000)*, pp. 169 -174, 2000.
- [4] J. Victory, C. C. McAndrew, J. Hall, and M. Zunino. A four-terminal compact model for high voltage diffused resistors with field plates. *IEEE Journal of Solid-State Circuits*, vol. 33, no. 9, pp. 1453 -1458, Sept. 1998.
- [5] *Verilog-AMS Language Reference Manual: Analog & Mixed-Signal Extensions to Verilog HDL*, version 2.1. Accellera, January 20, 2003. Available from www.accelera.com. An abridged version is available from www.verilog-ams.com or www.designers-guide.org/VerilogAMS.